Vinculando a Objetos

La sección previa demostró la vinculació de controles a SqlDataSource, que soporta propiedades para especificar cadenas de conexión, enunciados SQL o procedimientos almacenados utilizados para consultar y modificar la base de datos. Si bien esto es apropiado para la mayoría de sitios de pequeña escala, como sitios personales y de hobbyistas, el almacenar enunciados SQL directamente en las páginas de presentación de una aplicación puede rapidamente convertirse en difícil de mantener para aplicaciones a gran escala y a nivel empresarial. Estas aplicaciones típicamente requieren un modelo de datos más encapsulado usando una capa intermedia de acceso de datos o de componentes de negocios. Afortunadamente, el modelo de controles de fuente de datos de ASP.NET soporta este enfoque usando el control ObjectDataSource.
El modelo de objetos del control ObjectDataSource es similar al del control SqlDataSource. En vez de una propiedad ConnectionString, el ObjectDataSource expone una propiedad TypeName que especifica un tipo de objeto (nombre de clase) a ser instanciado, para llevar a cabo operaciones de datos. De forma similar a las propiedades de comando del SqlDataSource, el control ObjectDataSource soporta propiedades tales como SelectMethod, UpdateMethod, InsertMethod, y DeleteMethod para especificar métodos del tipo asociado que serán llamados para llevar a cabo estas operaciones de datos. Esta sección describe técnicas para construír una capa de acceso a datos y una capa de componentes de lógica de negocios, y luego exponerlos a través del control ObjectDataSource.
Vinculando a una Capa de Acceso a Datos
Una capa de componente de acceso a datos encapsula el código ADO.NET para consultar y modificar la base de datos a través de comandos SQL. Este típicamente abstrae los detalles de crear una conexión y comandos ADO.NET, y los expone a través de métodos que pueden ser llamados con los parámetros apropiados. Una capa de componente de acceso a datos típica podría estar expuesta como se muestra a continuación:

public class MyDataLayer {

  public DataView GetRecords();
  public DataView GetRecordsByCategory(String categoryName);
  public DataView GetRecordByID(int recordID);

  public int UpdateRecord(int recordID, String recordData);
  public int DeleteRecord(int recordID);
  public int InsertRecord(int recordID, String recordData);
}

El ObjectDataSource puede estar asociado a este tipo en la siguiente manera

<asp:ObjectDataSource TypeName="MyDataLayer" SelectMethod="GetRecords" UpdateMethod="UpdateRecord"     DeleteMethod="DeleteRecord" InsertMethod="InsertRecord" runat="server"/>

El ObjectDataSource requiere un patrón de diseño muy específico para los objetos con los cuales puede trabajar. Estas restricciones son impuestas en su mayoría por el ambiente carente de estado (stateless) bajo el cual se ejecutan las consultas (request) en una aplicación Web. Debido a que los objetos son típicamente creados y destruídos para servir cada consulta (request), los objetos vinculados a través de la fuente de datos de objeto, también se espera que sean carentes de estado (stateless) también. Por default, ObjectDataSource asume un constructor por default (sin argumentos) para el tipo especificado por la propiedad TypeName, aunque es posible instanciar este tipo en nombre del ObjectDataSource manejando el evento ObjectCreating, para crear un objeto customizado y asignarlo a la propiedad ObjectInstance de los argumentos del evento. El método de objeto asociado con la propiedad SelectMethod puede retornar cualquier objeto o lista Ienumerable, colección o arreglo (array). En el ejemplo de capa de acceso de datos mostrado arriba, el objeto DataView implementa la interfaz Ienumerable. Como discutiremos en la siguiene sección, esos métodos podrían también retornar una colección fuertemente tipeada (strongly typed) o un objeto.

GetProducts() -> ProductCollection
GetProductsDataSet() -> DataSet
GetProduct (int productId) -> Product

Los métodos Update (actualizar), Insert, y Delete deben tomar campos individuales de items de datos como parámetros, u opcionalmente pueden tomar  un objeto agregado de clase con propiedades públicas para los campos de items de datos.

UpdateProduct (int id, String name, double price, bool inStock)
UpdateProduct (Product p)  // p.ID, p.Name, p.Price, p.InStock 
DeleteProduct (int id)



Al igual que con el ejemplo con SqlDataSource, los nombres de parámetro o propiedades del item de datos pasado a los métodos de Update, Insert, y Delete deben concordar con los nombres de los campos de los items de datos retornados por el SelectMethod, para que funcionen los updates/deletes/inserts automáticos con los GridView/DetailsView. Al igual que con SqlDataSource, los parámetros para los métodos del ObjectDataSource pueden ser asociados con objetos de parámetros de datos a las colecciones SelectParameters, FilterParameters, UpdateParameters, DeleteParameters, o InsertParameters.
El siguiente ejemplo demuestra un control ObjectDataSource que expone datos de un componente de capa de acceso de datos nombrado AuthorsDB. El archivo de clase para este “tipo” está colocado en el directorio de aplicaciones App_Code, que es dinámicamente compilado por ASP.NET en el tiempo de corrida (runtime).

GridViewDAL_vb.aspx
<%@ Page Language="VB" %>
<html>

  <head runat="server">
    <title>GridView Bound to Data Access Layer</title>
  </head>

  <body>

    <form id="form1" runat="server">

      <asp:DropDownList ID="DropDownList1" Runat="server" DataSourceID="ObjectDataSource2"
        AutoPostBack="True" DataTextField="state" DataValueField="state" />

      <asp:ObjectDataSource ID="ObjectDataSource2" Runat="server" TypeName="AuthorsDB"
        SelectMethod="GetStates"/>

      <br />
      <br />

      <asp:GridView ID="GridView1" Runat="server" DataSourceID="ObjectDataSource1" AutoGenerateColumns="False"
        AllowPaging="True" AllowSorting="True" DataKeyNames="au_id">
        <Columns>
          <asp:CommandField ShowEditButton="True" />
          <asp:BoundField HeaderText="ID" DataField="au_id" SortExpression="au_id" ReadOnly="true" />
          <asp:BoundField HeaderText="Name" DataField="au_fname" SortExpression="au_fname" />
          <asp:BoundField HeaderText="LastName" DataField="au_lname" SortExpression="au_lname" />
          <asp:BoundField HeaderText="State" DataField="state" SortExpression="state" />
        </Columns>
      </asp:GridView>

      <asp:ObjectDataSource ID="ObjectDataSource1" Runat="server" TypeName="AuthorsDB"
        SelectMethod="GetAuthorsByState" UpdateMethod="UpdateAuthor" OldValuesParameterFormatString="{0}">
        <SelectParameters>
          <asp:ControlParameter Name="state" PropertyName="SelectedValue" ControlID="DropDownList1"/>
        </SelectParameters>
      </asp:ObjectDataSource>

    </form>

  </body>
</html>




App_Code/AuthorsDB.vb
'-----------------------------------------------------------------------
'  This file is part of the Microsoft .NET SDK Code Samples.
' 
'  Copyright (C) Microsoft Corporation.  All rights reserved.
' 
'This source code is intended only as a supplement to Microsoft
'Development Tools and/or on-line documentation.  See these other
'materials for detailed information regarding Microsoft code samples.
' 
'THIS CODE AND INFORMATION ARE PROVIDED AS IS WITHOUT WARRANTY OF ANY
'KIND, EITHER EXPRESSED OR IMPLIED, INCLUDING BUT NOT LIMITED TO THE
'IMPLIED WARRANTIES OF MERCHANTABILITY AND/OR FITNESS FOR A
'PARTICULAR PURPOSE.
'-----------------------------------------------------------------------
Imports System

Public Class AuthorsDB

    Public Sub New()
    End Sub

    Shared Function GetAuthorsByState(ByVal state As String) As System.Data.DataSet

        Dim connectionString As String = ConfigurationManager.ConnectionStrings("Pubs").ConnectionString
        Dim dbConnection As System.Data.IDbConnection = New System.Data.SqlClient.SqlConnection(connectionString)

        Dim queryString As String

    If Not (state = String.Empty) Then
        queryString= "SELECT [authors].[au_id], [authors].[au_fname], [authors].[au_lname], [authors].[" & _
                "state] FROM [authors] WHERE ([authors].[state] = @state)"
    Else
        queryString= "SELECT [authors].[au_id], [authors].[au_fname], [authors].[au_lname], [authors].[" & _
                "state] FROM [authors]"
    End If

        Dim dbCommand As System.Data.IDbCommand = New System.Data.SqlClient.SqlCommand
        dbCommand.CommandText = queryString
        dbCommand.Connection = dbConnection

    If Not (state = String.Empty) Then
            Dim dbParam_state As System.Data.IDataParameter = New System.Data.SqlClient.SqlParameter
            dbParam_state.ParameterName = "@state"
            dbParam_state.Value = state
            dbParam_state.DbType = System.Data.DbType.StringFixedLength
            dbCommand.Parameters.Add(dbParam_state)
    End If






        Dim dataAdapter As System.Data.IDbDataAdapter = New System.Data.SqlClient.SqlDataAdapter

        dataAdapter.SelectCommand = dbCommand

        Dim dataSet As System.Data.DataSet = New System.Data.DataSet
        dataAdapter.Fill(dataSet)

        Return dataSet

    End Function

    Shared Function GetStates() As System.Data.DataSet

        Dim connectionString As String = ConfigurationManager.ConnectionStrings("Pubs").ConnectionString
        Dim dbConnection As System.Data.IDbConnection = New System.Data.SqlClient.SqlConnection(connectionString)

        Dim queryString As String = "SELECT DISTINCT [authors].[state] FROM [authors]"
        Dim dbCommand As System.Data.IDbCommand = New System.Data.SqlClient.SqlCommand

        dbCommand.CommandText = queryString
        dbCommand.Connection = dbConnection

        Dim dataAdapter As System.Data.IDbDataAdapter = New System.Data.SqlClient.SqlDataAdapter

        dataAdapter.SelectCommand = dbCommand

        Dim dataSet As System.Data.DataSet = New System.Data.DataSet
        dataAdapter.Fill(dataSet)

        Return dataSet

    End Function

    Shared Function UpdateAuthor(ByVal au_id As String, ByVal au_lname As String, ByVal au_fname As String, ByVal state As String) As Integer

        Dim connectionString As String = ConfigurationManager.ConnectionStrings("Pubs").ConnectionString
        Dim dbConnection As System.Data.IDbConnection = New System.Data.SqlClient.SqlConnection(connectionString)

        Dim queryString As String = "UPDATE [authors] SET [au_lname]=@au_lname, [au_fname]=@au_fname, [state]=@state W" & _
            "HERE ([authors].[au_id] = @au_id)"
        Dim dbCommand As System.Data.IDbCommand = New System.Data.SqlClient.SqlCommand

        dbCommand.CommandText = queryString
        dbCommand.Connection = dbConnection

        Dim dbParam_au_id As System.Data.IDataParameter = New System.Data.SqlClient.SqlParameter
        dbParam_au_id.ParameterName = "@au_id"
        dbParam_au_id.Value = au_id
        dbParam_au_id.DbType = System.Data.DbType.[String]
        dbCommand.Parameters.Add(dbParam_au_id)
        Dim dbParam_au_lname As System.Data.IDataParameter = New System.Data.SqlClient.SqlParameter
        dbParam_au_lname.ParameterName = "@au_lname"
        dbParam_au_lname.Value = au_lname
        dbParam_au_lname.DbType = System.Data.DbType.[String]
        dbCommand.Parameters.Add(dbParam_au_lname)
        Dim dbParam_au_fname As System.Data.IDataParameter = New System.Data.SqlClient.SqlParameter
        dbParam_au_fname.ParameterName = "@au_fname"
        dbParam_au_fname.Value = au_fname
        dbParam_au_fname.DbType = System.Data.DbType.[String]
        dbCommand.Parameters.Add(dbParam_au_fname)
        Dim dbParam_state As System.Data.IDataParameter = New System.Data.SqlClient.SqlParameter
        dbParam_state.ParameterName = "@state"
        dbParam_state.Value = state
        dbParam_state.DbType = System.Data.DbType.StringFixedLength
        dbCommand.Parameters.Add(dbParam_state)

        Dim rowsAffected As Integer = 0

        dbConnection.Open()

        Try
            rowsAffected = dbCommand.ExecuteNonQuery
        Finally
            dbConnection.Close()
        End Try

        Return rowsAffected

    End Function

End Class

Vinculando a una Capa de Lógica de Negocios
Una cosa importante que hay que notar sobre una capa de acceso a datos es que debido a que el SelectMethod retorna el resultado de ejecutar la consulta como un DataView, aún así expone el Schema de la base de datos subyacente a las páginas de presentación. Otra cosa a notar es que no hay reglas de negocio en la capa de acceso a datos; simplemente ejecuta consultas y retorna resultados. Para aislar la presentación del esquema de la base de datos e introducir reglas de negocio o validaciones, la capa de acceso a datos es típicamente envuleto en una capa de lógica de negocios.
Una capa de lógica de negocios es similar a un DAL (Data Access Logic) en que expone métodos al ObjectDataSource para vincular controles en páginas Web. Sin embargo, en vez de retornar resultados de ADO.NET directamente, típicamente retorna objetos fuertemente tipeados (strongly-typed) que representan las entidades de negocio usados por la aplicación. Esto desconecta (decouples) la capa de presentación del esquema del almacenamiento de datos subyacente, facilitando mantener la porción de acceso a datos de su sitio separado de las páginas que consumen los datos. Con una capa intermedia apropiadamente arquitecturada, Ud. podría cambiar el almacenamiento de datos subyacente a un esquema completamente diferente sin tener que actualizar las páginas individuales en su aplicación.
Un ejemplo de una capa de lógica de negocios es presentada abajo.
public class MyBusinessLayer {

  public RecordCollection GetRecords();
  public RecordCollection GetRecordsByCategory(String categoryName);
  public RecordCollection GetRecordByID(int recordID);
  public String GetRecordName(int recordID);
  public Object GetRecordData(int recordID);

  public int UpdateRecord(Record r);
  public int DeleteRecord(Record r);
  public int InsertRecord(Record r);

  public int UpdateRecordData(int ID, String Data);
  public int UpdateRecordName(int ID, String Name);
}

public class Record {
  public int ID { get; set; }
  public String Name { get; set; }
  public Object Data { get; set; }
}

La diferencia principal entre la capa de lógica de negocios y la capa de acceso de datos es que retorna un RecordCollection fuertemente tipeado (strongly typed) de objetos Record, en vez de un DataView. También permite operaciones de Update, Insert, y Delete que toman este objeto Record como parámetro. La propiedad DataObjectTypeName del ObjectDataSource permite que Ud. configure el ObjectDataSource para que pase este tipo a los métodos, en vez de valores de campo individuales. En la implementación de métodos de la capa de lógica de negocios, Ud. puede incluír lógica customizada para validar las reglas de negocio. Por ejemplo, Ud. puede asegurarse que sólo registros en la categoría “En Revisión” son actualizados, o sólo un Administrator puede insertar nuevos registros. Ud. también puede incluír lógica de validación para asegurarse de que los tipos de datos y valores suplidos como argumentos estén correctos antes de insertar o modificar datos en la base de datos. Note que la lógica de validación en la capa de negocios no es un sustituto para la validación de entradas en la capa de presentación, que ayudan a guiar al usuario final para que ingrese los valores correctos antes de enviar la actualización.
El ejemplo abajo muestra una capa de lógica de negocios simple, llamada AuthorsComponent. Internamente, este BLL (Business Logic Layer) hace llamados a través del DAL (Data Access Logic) para llevar a cabo las operaciones de base de datos. Por simplicidad, este BLL no incluye ninguna regla de negocios o validación, aunque en una aplicación de la vida real sí los incluiría. También note que en vez de escribir nuestra propia clase de collección customizada para retornar registros fuertemente tipeados (strongly- typed), este ejemplo toma ventaja de una nueva característica de lenguaje en el .NET Framework 2.0 llamado “Generics” para crear una colección de objetos Author. Al usar colecciones fuertemente tipeados le permite al ObjectDataSource inferir el esquema de su objeto de negocios en tiempo de diseño (en Visual Studio y otras herramientas).

GridViewObject_vb.aspx
<%@ Page Language="VB" %>
<html>

  <head id="Head1" runat="server">
    <title>GridView Bound to Data Component</title>
  </head>

  <body>

    <form id="form1" runat="server">

      <asp:DropDownList ID="DropDownList1" Runat="server" DataSourceID="ObjectDataSource2"
        AutoPostBack="True" />

      <asp:ObjectDataSource ID="ObjectDataSource2" Runat="server" TypeName="AuthorsComponent"
        SelectMethod="GetStates"/>

      <br />
      <br />

      <asp:GridView ID="GridView1" Runat="server" DataSourceID="ObjectDataSource1" AutoGenerateColumns="False"
        AllowPaging="True" AllowSorting="True">
        <Columns>
          <asp:CommandField ShowEditButton="True" />
          <asp:BoundField HeaderText="ID" DataField="ID" SortExpression="ID" />
          <asp:BoundField HeaderText="Name" DataField="Name" SortExpression="Name" />
          <asp:BoundField HeaderText="LastName" DataField="LastName" SortExpression="LastName" />
          <asp:BoundField HeaderText="State" DataField="State" SortExpression="State" />
        </Columns>
      </asp:GridView>

      <asp:ObjectDataSource ID="ObjectDataSource1" Runat="server" TypeName="AuthorsComponent"
        SelectMethod="GetAuthorsByState" UpdateMethod="UpdateAuthor" DataObjectTypeName="Author" SortParameterName="sortExpression">
        <SelectParameters>
          <asp:ControlParameter Name="state" PropertyName="SelectedValue" ControlID="DropDownList1"></asp:ControlParameter>
        </SelectParameters>
      </asp:ObjectDataSource>

    </form>
  </body>
</html>

 
App_Code/Author.vb
'-----------------------------------------------------------------------
'  This file is part of the Microsoft .NET SDK Code Samples.
' 
'  Copyright (C) Microsoft Corporation.  All rights reserved.
' 
'This source code is intended only as a supplement to Microsoft
'Development Tools and/or on-line documentation.  See these other
'materials for detailed information regarding Microsoft code samples.
' 
'THIS CODE AND INFORMATION ARE PROVIDED AS IS WITHOUT WARRANTY OF ANY
'KIND, EITHER EXPRESSED OR IMPLIED, INCLUDING BUT NOT LIMITED TO THE
'IMPLIED WARRANTIES OF MERCHANTABILITY AND/OR FITNESS FOR A
'PARTICULAR PURPOSE.
'-----------------------------------------------------------------------
Imports System

Public Class Author

    Private _id As String

    Public Property ID() As String
        Get
            Return _id
        End Get
        Set(ByVal value As String)
            _id = value
        End Set
    End Property

    Private _name As String

    Public Property Name() As String
        Get
            Return _name
        End Get
        Set(ByVal value As String)
            _name = value
        End Set
    End Property

    Private _lastName As String

    Public Property LastName() As String
        Get
            Return _lastName
        End Get
        Set(ByVal value As String)
            _lastName = value
        End Set
    End Property




    Private _state As String

    Public Property State() As String
        Get
            Return _state
        End Get
        Set(ByVal value As String)
            _state = value
        End Set
    End Property

    Public Sub New(ByVal id As String, ByVal name As String, ByVal lastName As String, ByVal state As String)
        Me.ID = id
        Me.Name = name
        Me.LastName = lastName
        Me.State = state
    End Sub

    Public Sub New()
        ' Default Constructor
    End Sub

End Class







App_Code/AuthorsComponent.vb
'-----------------------------------------------------------------------
'  This file is part of the Microsoft .NET SDK Code Samples.
' 
'  Copyright (C) Microsoft Corporation.  All rights reserved.
' 
'This source code is intended only as a supplement to Microsoft
'Development Tools and/or on-line documentation.  See these other
'materials for detailed information regarding Microsoft code samples.
' 
'THIS CODE AND INFORMATION ARE PROVIDED AS IS WITHOUT WARRANTY OF ANY
'KIND, EITHER EXPRESSED OR IMPLIED, INCLUDING BUT NOT LIMITED TO THE
'IMPLIED WARRANTIES OF MERCHANTABILITY AND/OR FITNESS FOR A
'PARTICULAR PURPOSE.
'-----------------------------------------------------------------------
Imports System
Imports System.Data
Imports System.Collections.Generic

Public Class AuthorsComponent

    Public Sub New()

        ' TODO: Add constructor logic here
    End Sub

    Public Function GetAuthorsByState(ByVal state As String, ByVal sortExpression As String) As List(Of Author)

        Dim authors As New List(Of Author)

        Dim ds As DataSet = AuthorsDB.GetAuthorsByState(state)

        Dim row As DataRow
        For Each row In ds.Tables(0).Rows
            authors.Add(New Author(row("au_id"), row("au_fname"), row("au_lname"), row("state")))
        Next

        authors.Sort(New AuthorComparer(sortExpression))
        Return authors
    End Function

    Public Function UpdateAuthor(ByVal ID As String, ByVal LastName As String, ByVal Name As String, ByVal State As String) As Integer

        Return AuthorsDB.UpdateAuthor(ID, LastName, Name, State)
    End Function

    Public Function UpdateAuthor(ByVal a As Author) As Integer

        Return AuthorsDB.UpdateAuthor(a.ID, a.LastName, a.Name, a.State)
    End Function




    Public Function GetStates() As List(Of String)

        Dim states As New List(Of String)
        Dim ds As DataSet = AuthorsDB.GetStates()

        Dim row As DataRow
        For Each row In ds.Tables(0).Rows

            states.Add(row("state"))
        Next
        Return states
    End Function

End Class

Public Class AuthorComparer : Implements IComparer(Of Author)

    Private _sortColumn As String
    Private _reverse As Boolean

    Public Sub New(ByVal sortExpression As String)

        _reverse = sortExpression.ToLowerInvariant().EndsWith(" desc")
        If (_reverse) Then

            _sortColumn = sortExpression.Substring(0, sortExpression.Length - 5)
        Else

            _sortColumn = sortExpression
        End If
    End Sub

    Public Function IComparer_Compare(ByVal a As Author, ByVal b As Author) As Integer Implements IComparer(Of Author).Compare

        Dim retVal As Integer = 0
        Select Case _sortColumn

            Case "ID"
                retVal = String.Compare(a.ID, b.ID, StringComparison.InvariantCultureIgnoreCase)
            Case "Name"
                retVal = String.Compare(a.Name, b.Name, StringComparison.InvariantCultureIgnoreCase)
            Case "LastName"
                retVal = String.Compare(a.LastName, b.LastName, StringComparison.InvariantCultureIgnoreCase)
            Case "State"
                retVal = String.Compare(a.State, b.State, StringComparison.InvariantCultureIgnoreCase)
        End Select

        Dim _reverseInt As Integer = -1
        If (_reverse) Then
            _reverseInt = 1
        End If
        Return (retVal * _reverseInt)
    End Function

End Class


App_Code/AuthorsDB.vb
'-----------------------------------------------------------------------
'  This file is part of the Microsoft .NET SDK Code Samples.
' 
'  Copyright (C) Microsoft Corporation.  All rights reserved.
' 
'This source code is intended only as a supplement to Microsoft
'Development Tools and/or on-line documentation.  See these other
'materials for detailed information regarding Microsoft code samples.
' 
'THIS CODE AND INFORMATION ARE PROVIDED AS IS WITHOUT WARRANTY OF ANY
'KIND, EITHER EXPRESSED OR IMPLIED, INCLUDING BUT NOT LIMITED TO THE
'IMPLIED WARRANTIES OF MERCHANTABILITY AND/OR FITNESS FOR A
'PARTICULAR PURPOSE.
'-----------------------------------------------------------------------
Imports System
Imports System.Data
Imports System.Collections.Generic

Public Class AuthorsComponent

    Public Sub New()

        ' TODO: Add constructor logic here
    End Sub

    Public Function GetAuthorsByState(ByVal state As String, ByVal sortExpression As String) As List(Of Author)

        Dim authors As New List(Of Author)

        Dim ds As DataSet = AuthorsDB.GetAuthorsByState(state)

        Dim row As DataRow
        For Each row In ds.Tables(0).Rows
            authors.Add(New Author(row("au_id"), row("au_fname"), row("au_lname"), row("state")))
        Next

        authors.Sort(New AuthorComparer(sortExpression))
        Return authors
    End Function

    Public Function UpdateAuthor(ByVal ID As String, ByVal LastName As String, ByVal Name As String, ByVal State As String) As Integer

        Return AuthorsDB.UpdateAuthor(ID, LastName, Name, State)
    End Function

    Public Function UpdateAuthor(ByVal a As Author) As Integer

        Return AuthorsDB.UpdateAuthor(a.ID, a.LastName, a.Name, a.State)
    End Function



    Public Function GetStates() As List(Of String)

        Dim states As New List(Of String)
        Dim ds As DataSet = AuthorsDB.GetStates()

        Dim row As DataRow
        For Each row In ds.Tables(0).Rows

            states.Add(row("state"))
        Next
        Return states
    End Function

End Class

Public Class AuthorComparer : Implements IComparer(Of Author)

    Private _sortColumn As String
    Private _reverse As Boolean

    Public Sub New(ByVal sortExpression As String)

        _reverse = sortExpression.ToLowerInvariant().EndsWith(" desc")
        If (_reverse) Then

            _sortColumn = sortExpression.Substring(0, sortExpression.Length - 5)
        Else

            _sortColumn = sortExpression
        End If
    End Sub

    Public Function IComparer_Compare(ByVal a As Author, ByVal b As Author) As Integer Implements IComparer(Of Author).Compare

        Dim retVal As Integer = 0
        Select Case _sortColumn

            Case "ID"
                retVal = String.Compare(a.ID, b.ID, StringComparison.InvariantCultureIgnoreCase)
            Case "Name"
                retVal = String.Compare(a.Name, b.Name, StringComparison.InvariantCultureIgnoreCase)
            Case "LastName"
                retVal = String.Compare(a.LastName, b.LastName, StringComparison.InvariantCultureIgnoreCase)
            Case "State"
                retVal = String.Compare(a.State, b.State, StringComparison.InvariantCultureIgnoreCase)
        End Select

        Dim _reverseInt As Integer = -1
        If (_reverse) Then
            _reverseInt = 1
        End If
        Return (retVal * _reverseInt)
    End Function

End Class


La ilustración abajo muestra la interacción entre GridView, ObjectDataSource, y la capa de lógica de negocios. El ObjectDataSource está configurado para llamar el método GetContacts en el tipo ContactsList, que retorna una colección de objetos tipo Contact. El GridView enumera estos objetos Contacts y los vincula directamente a las propiedades (ID, Name) de este tipo para crear las columnas. Note que el SelectMethod puede retornar tanto un IEnumerable de objetos Contact, o puede retor un objeto Contact singleton. El ObjectDataSource siempre envolverá los resultados del SelectMethod en un IEnumerable si no implementa su propio IEnumerable.
[image: ]
Al igual que SqlDataSource, el control ObjectDataSource soporta ordenamiento (sort) cuando el SelectMethod retorna un objeto DataSet, DataView, o DataTable. Internamente, el ObjectDataSource se apoya en la propiedad DataView.Sort para llevar a cabo el ordenamiento para estos casos. ObjectDataSource también soporta ordenamiento customizado en la implementación del SelectMethod, lo que es útil si el método no retorna un DataSet, DataView, o DataTable. El ordenamiento customizado se configura especificando la propiedad SortParameterName para que tenga el nombre de un parámetro de método que acepte el SortExpression de la fuente de datos. Cuando llama al SelectMethod, el ObjectDataSource pasará esta expresión a su método, y Ud. puede implementar su propia lógica de ordenamiento usando esta expresión. El ejemplo precedente demuestra una implementación customicada de ordenamiento en la clase AuthorsComponent.
El ObjectDataSource también soporta paginación customizada en la implementación del SelectMethod. Esto se configura usando las propiedades StartRowIndexParameterName, MaximumRowsParameterName, y SelectCountMethod, y se discute en más detalle en la sección “Advanced Paging and Sorting” de este tutorial.



Vinculando a un DataSet de Visual Studio
El construír una capa de acceso de datos puede ser tedioso, ya que frecuentemente el código ADO.NET para ejecutar enunciados SQL o procedimientos almacenados es el mismo o similar para los diferentes métodos del DAL. A pesar de que Ud. puede escribir su propio DAL usando código ADO.NET customizado usando las técnicas de arriba, Visual Studio  también provee una forma conveniente de generarle una capa de acceso de datos, basado en sus entradas a un asistente (wizard) sencillo. En este caso la capa de acceso a datos es un objeto DataSet fuertemente tipeado (strongly-typed). El DataSet contiene tipos TableAdapter que exponen métodos para retornar objetos DataTable fuertemente tipeados. Estos métodos son apropiados para vincularse a un ObjectDataSource directamente, o para llamarlos desde su componente de capa de lógica de negocios.
Para adicionar un DataSet a su proyecto Visual Studio, dé click derecho en el Solution Explorer y escogió “Add New Item…” y luego seleccione el tipo de item de proyecto “DataSet”. Visual Studio puede preguntarle por adicionar esto al directorio App_Code. Conteste “Yes” (Sí) a esta pregunta para proceder. Visual Studio adiciona un archivo DataSet.xsd al directorio App_Code y abre el diseñador de DataSets, lanzando el asistente (wizard) del TableAdapter por default. Al ir por los pasos del asistente (wizard) del TableAdapter, Ud. puede especificar enunciados SQL o procedimientos almacenados de su base de datos, y luego nombrar los métodos asociados a estas consultas/comandos en la página final del asistente (wizard).
[image: ]
El TableAdapter puede exponer dos tipos de métodos: métodos Fill que son apropiados para llenar un DataSet existente, y métodos Get que retornan un objeto DataTable ya llenado. El primero es más apropiado para un cliente Windows (en donde el DataSet es mantenido en memoria por el tiempo de vida de la aplicación), meintras que los métodos Get son más apropiados para el ObjectDataSource. El asistente (wizard) del TableAdapter también puede generar automáticamente operaciones de Select, Update, Insert o Delete sobre dicho esquema. Ud. puede agregar múltiples TableAdapters en un DataSet al dar click derecho en el diseñador de DataSet. También puede agregar consultas (queries) adicionales al TableAdapter dando click derecho a la caja del TableAdapter en el diseñador, siempre que retornen el mismo schema (lista de campos). Por ejemplo, Ud. podría tener un solo TableAdapter con los métodos GetAuthors() y GetAuthorsById(int id), pero necesitará un nuevo TableAdapter adicional para tener un método GetTitles(). La figura abajo muestra el diseñador de DataSet con varios TableAdapters adicionados:
[image: ]
Una vez que haya terminado el diseño del DataSet, almacene el archivo DataSet.xsd (esto causa que los tipos sean compilados  por el diseñador en segundo plano, para que estén disponibles para sus páginas). Ud. puede ahora ver estos tipos expuestos al código de la página:

protected void Page_Load(object sender, EventArgs e)
{
    DataSetTableAdapters.PhotosTableAdapter adapter = new DataSetTableAdapters.PhotosTableAdapter();
    adapter.GetPhotosForAlbum(0);
}



Sin embargo, Ud. no necesita llamar estos métodos directamente desde su propio código. En vez de ello, Ud. puede vincular el ObjectDataSource a estos métodos.

<asp:ObjectDataSource ID="ObjectDataSource1" runat="server"
  TypeName="DataSetTableAdapters.PhotosTableAdapter"
  SelectMethod="GetPhotosForAlbum">
  <SelectParameters>
    <asp:QueryStringParameter Name="albumID" QueryStringField="id" Type="Int32"/>
  </SelectParameters>
</asp:ObjectDataSource>

El siguiente ejemplo muestra un ObjectDataSource vinculado a los métodos del DataSet.TableAdapter. Utilizaremos este DataSet en varios otros ejemplos posteriores en el tutorial, para demostrar cómo puede utilizar controles de datos ASP.NET para implementar una aplicación simple de álbum de fotografías. Note que el DetailsView en este ejemplo está usando un nuevo tipo de campo llamado ImageField para desplegar las imágenes fotográficas. También note el uso del ConvertNullToDBNull en el ObjectDataSource para que los valores de parámetro nulos sean convertidos a DbNull antes que que sean pasados a los métodos del TableAdapter (requerido).
VB Vinculando a un DataSet del Visual Studio

AlbumsDataSet.aspx
<%@ Page Language="VB" StylesheetTheme="White"  %>
<!DOCTYPE html PUBLIC "-//W3C//DTD XHTML 1.0 Transitional//EN" "http://www.w3.org/TR/xhtml1/DTD/xhtml1-transitional.dtd">

<html xmlns="http://www.w3.org/1999/xhtml">

<head runat="server">
    <title>Binding to a Visual Studio DataSet</title>
</head>

<body>

    <form id="form1" runat="server">
        <div>
            <h1>
                My Photo Album Site</h1>
            <b>Select an owner:

                <asp:DropDownList ID="DropDownList1" runat="server" DataSourceID="ObjectDataSource2"
                    DataTextField="Owner" DataValueField="Owner" AutoPostBack="True" /></b>

            <br />
            <br />

            <asp:GridView ID="GridView1" runat="server" DataSourceID="ObjectDataSource1" AutoGenerateColumns="False"
                AllowPaging="True" AllowSorting="True" DataKeyNames="AlbumID">
                <Columns>
                    <asp:CommandField ShowEditButton="True" />
                    <asp:BoundField ReadOnly="True" HeaderText="AlbumID" DataField="AlbumID" SortExpression="AlbumID" />
                    <asp:BoundField HeaderText="AlbumName" DataField="AlbumName" SortExpression="AlbumName" />
                    <asp:BoundField HeaderText="Owner" DataField="Owner" SortExpression="Owner" />
                    <asp:BoundField HeaderText="DateCreated" DataField="DateCreated" SortExpression="DateCreated" />
                    <asp:BoundField HeaderText="AlbumPath" DataField="AlbumPath" SortExpression="AlbumPath" />
                    <asp:CheckBoxField HeaderText="Visible" SortExpression="Visible" DataField="Visible" />
                    <asp:HyperLinkField HeaderText="View Photos..." Text="View Photos..." DataNavigateUrlFields="AlbumID,AlbumName"
                        DataNavigateUrlFormatString="PhotosDetailsView.aspx?ID={0}&amp;Name={1}" />
                </Columns>
            </asp:GridView>

            <br />




            <asp:ObjectDataSource ID="ObjectDataSource1" runat="server" TypeName="DataComponentTableAdapters.AlbumsTableAdapter"
                SelectMethod="GetAlbumsByOwner" UpdateMethod="Update" ConvertNullToDBNull="true" OldValuesParameterFormatString="original_{0}">
                <SelectParameters>
                    <asp:ControlParameter ControlID="DropDownList1" Name="Owner" PropertyName="SelectedValue"
                        Type="String" />
                </SelectParameters>
                <UpdateParameters>
                    <asp:Parameter Name="AlbumName" Type="String" />
                    <asp:Parameter Name="Owner" Type="String" />
                    <asp:Parameter Name="DateCreated" Type="DateTime" />
                    <asp:Parameter Name="AlbumPath" Type="String" />
                    <asp:Parameter Name="Visible" Type="Boolean" />
                    <asp:Parameter Name="Original_AlbumID" Type="Int32" />
                </UpdateParameters>
            </asp:ObjectDataSource>

            <asp:ObjectDataSource ID="ObjectDataSource2" runat="server" TypeName="DataComponentTableAdapters.OwnersTableAdapter"
                SelectMethod="GetOwners" />

        </div>
    </form>
</body>
</html>



PhotosDetailsView.aspx
<%@ Page Language="VB" StylesheetTheme="White" %>
<!DOCTYPE html PUBLIC "-//W3C//DTD XHTML 1.0 Transitional//EN" "http://www.w3.org/TR/xhtml1/DTD/xhtml1-transitional.dtd">
<html xmlns="http://www.w3.org/1999/xhtml">

<head runat="server">
  <title>View Photos</title>
</head>

<body>

  <form id="form1" runat="server">
    <div>
      <h1>
        Album: <%= Server.HtmlEncode(Request.QueryString("Name")) %></h1>

      <asp:DetailsView AutoGenerateRows="False" DataKeyNames="PhotoID" DataSourceID="ObjectDataSource1"
        HeaderText="Album Photos" Height="50px" ID="DetailsView1" runat="server" Width="125px">
        <Fields>
          <asp:BoundField DataField="Caption" HeaderText="Caption" ShowHeader="False" SortExpression="Caption" />
          <asp:ImageField DataImageUrlField="FileName" DataImageUrlFormatString="~/Images/{0}"
            NullDisplayText="No Image to Display" ReadOnly="True" ShowHeader="False">
          </asp:ImageField>
        </Fields>
      </asp:DetailsView>

      <br />
      <br />

      <asp:ObjectDataSource ID="ObjectDataSource1" runat="server" TypeName="DataComponentTableAdapters.PhotosTableAdapter"
        SelectMethod="GetPhotosForAlbum">
        <SelectParameters>
          <asp:QueryStringParameter Name="AlbumID" QueryStringField="ID" DefaultValue="1" />
        </SelectParameters>
      </asp:ObjectDataSource>

    </div>
  </form>
</body>
</html>






DataComponent.xsd
[image: ]

